BEAM HEADINGS AND QTH LOCATORS ON YOUR MICRO

By Greg Baker

The LOCATOR program is a dual purpose program combining a QTH Locator program and a Great Circle program. The program demands as input either (a) the QTH Locator, or (b) the latitude and longitude of the target station.

If the QTH Locator is provided as an input, the program calculates latitude and longitude of the centre of the locator square then the great circle bearing and path distances. If the latitude and longitude of the target station are input, the program calculates the QTH Locator square then the great circle bearings and path distances.

The program has been written for and tested on an unexpanded Dick Smith VZ-200 computer. The entire program is written in BASIC and should be adaptable to most BASIC versions.

QTH Locators

QTH Locators are an alternative to the use of latitude and longitude for specifying the location of amateur radio stations around the world. For this purpose, the earth's surface is first divided into 18 x 18 = 324 fields, each 20 degrees wide in longitude and 10 degrees wide in latitude.

Each of these fields is then divided into 10 x 10 = 100 squares, each 2 degrees wide in longitude and 1 degree wide in latitude. These squares are further sub-divided into 24 x 24 = 576 sub-squares of 5 minutes longitude by 2.5 minutes latitude. Figure 1 shows how these fields, squares and sub-squares are labeled.

From these labels, a six-character QTH Locator is formed. Note that the two character field, square and sub-square labels are longitude first, latitude second, and are labeled consecutively from west to east for longitude and south to north for latitude.

The full six character locator has the form f1f2d1d2s1s2 where f1f2 is the alpha field locator, d1d2 is the numeric square locator, and s1s2 is the alpha sub-square locator. For example, the author's QTH is at 35°24.4' South latitude by 149°57.3' East longitude, which corresponds to a QTH Locator of QF44XO.

It is not necessary to always use the six character QTH Locator. If a coarser grid with less accuracy is satisfactory, the first four character's can be used. For less accuracy again, use just the first two characters. Further details of the QTH Locator system can be found in Tony Gilbert's 'Traffic' column, ARA Vol 7, No 9, Page 5.

Great Circle Bearings And Distances

Great Circle bearings are the true bearings for beam aiming. Due to the curvature of the earth, bearings obtained from standard (mercator projection) maps are not accuracte over more than a few degrees. Two bearings 180° apart are usually given — the short path bearing and the long path bearing. Similarly, there are two Great Circle distances — that for the short path and that for the long path.



For more details on Great Circle bearings, see articles in ARA Vol 6, No 9, and ARA Vol 7, No 2, both available from ARA Reprints (Back Issues Department).

Flowchart and Algorithms

Unlike some other locator programs, the main calculations used here are neat and compact. The program incorporates extensive error checking, which is good for the VZ-200 but may not work on other systems.

Because the calculations are complex, great care should be taken to type them is correctly. Statements to be particularly careful with are those in lines 390, 400, 510 and 520.

The program flowchart is shown in Figure 2.

Originating Station

The program as it is written incorporates the latitude and longitude of Mount Ainslie, Canberra, as the location of the station from which the bearings are calculated. To function correctly from any other location, latitude and longitude for that QTH need to be inserted at lines 100 and 110 respectively.

Minutes of arc should be divided by 60 and added to the degrees. Seconds of arc should be divided by 3600 and added to the degree to give a decimalised latitude and longitude. Then the latitude and longitude should be give a sign — positive for north latitudes and east longitude; negative for south latitudes and west longitudes.

1 of 5. ? Apr 87



Fig. 1 — How the QTH Locator system is calculated.

For example, a station at 33°55' South by 151°10' East has a decimalised latitude of -33 + 55/60 = -33.91667 and longitude of +151 + 10/60 = +151.16667.

Alternatively, because the program allows the origin to be changed while it is running — for use away from the normal QTH for example — the user could type in their own QTH every time the program is run, although it would be easier to make the change permanent. Final output prints a new origin reminder message if this option has been exercised.

Using The Program

On running the program, the user is asked whether s/he wants to alter the latitude and longitude of their station. Enter 'Y' to choose this option or any other character to bypass it. If 'Y' is selected, you will be asked to enter the new decimalised latitude and longitude of origin.

If a valid latitude and longitude is entered, the program proceeds. Otherwise an error message is displayed for a short period and the user is requested to re-enter the origin coordinates.

Next, the program requests the target QTH name, followed by the option to enter the QTH Locator or the latitude and longitude of the location. The target name is truncated to 22 characters after entry and further truncated to nine characters if the new origin option is chosen to allow room on the printout for the new origin reminder message.

If the user chooses to enter a QTH Locator, a valid two, four or six character locator must be entered before the program will proceed to the Great Circle calculations which will use the latitude and longitude of the locator field, square or sub-square centre as the target location.

Similarly latitude and longitude, if entered, must be valid before the program will proceed.

Once great circle bearings and distances are calculated, the program prints results and asks the user to enter another target.

A few typical outputs are shown in photographs accompanying this article.

Warnings

The great circle section of the program produces errors if the target is within 50 kilometres of the origin station (when it wouldn't be usual to use a great circle program anyway), or if the target is close to either the north or south pole (although, again, it wouldn't be usual to use a great circle program to point your beam due north or south anyway).

Note that ARA Vol 9, No 4, has an article on short range beam headings for VHF and UHF enthusiasts.

Test Data

Table 1 shows program output data for the origin station located at 35°16' South, 149° East as incorporated in program statements at lines 100 and 110. This test data shold be used to check the program before the data in lines 100 and 110 is changed for your QTH.

Copies of the program for VZ-200 can be obtained on cassette from the author for $7.00 post-paid. Write to Greg Baker, PO Box 93, Braidwood, NSW 2622. Comments and suggestions (with an SASE for reply) can be sent to the same address.

Debugged disk copies of the program modified for **Commodore VIC-20, C-64 or C-128** can be obtained by sending $10 or a blank formatted disk and $5 (includes postage) to High-Tech Media, 4 Renshaw St, Doncaster East 3109.

Diagram 2. Flowchart



	Target: Name / Latitude	QTH Locator: / Longitude	Short Path: True Bearing	Short Path: Distance
1.	Buffalo	FN02MU	63°26'	15826 km
	42°52'N	78°55'W		
2.	Hong Kong	OL72CF	324°40'	7374
	22°15'N	114°15'E		
3.	Falklands	GD08FL	162°37'	9963
	51°30'S	59°30'W		
4.	Auckland	RF73JB	102°07'	2301
	36°55'S	174°47'S		

PROGRAM LISTING FOR VZ-200

```
0010 REM PROGRAM "LOCATOR"
0020 REM GREG BAKER, BRAIDWOOD, 2622
0030 DIM C(6),CB(3),CM(3),CT(3),L(2,3),M(6),N$(2),S(2),
T(2,2),TG(2)
0035 DIM F$(2),G$(2),H%(2)
0040 DATA 65,82,10,48,57,1,65,88,0.041667
0045 DATA "NORTH","EAST","SOUTH","WEST"
0047 W$=" "
0050 FOR I=1 TO 3
0060 READ CB(I),CT(I),CM(I)
0070 NEXT I
0072 FOR I=1 TO 2
0074 READ F$(I),G$(I)
0076 NEXT I
0080 REM ORIGIN STATION LAT/LONG
0090 REM INSERT YOUR OWN QTH HERE
0100 T(1,1)=-35.2667
0110 T(2,1)=149.1667
0120 CLS
0122 PRINT"ENTER 'Y' TO ALTER ORIGIN";:INPUT Y$
0130 IF Y$<>"Y" THEN 170
0140 PRINT@192,"NEW ORIGIN LAT/LONG"
0145 C$=" ★NEW ORIGIN★"
0150 K=1
```


3 of 5.





```
0160 GOSUB 1030
0170 CLS
0175 PRINT"TARGET NAME";:INPUT T$
0180 T$=LEFT$(T$,22)
0190 FL=0
0200 PRINT@64,"ENTER: '1' FOR TARGET QTH
LOCATOR"
0210 INPUT" '2' FOR TARGET LAT/LONG";Y
0220 if Y=2 THEN 420
0230 IF Y=1 THEN 240
0235 PRINT@152," ":GOTO 200
0240 PRINT@192,"LOCATOR";:INPUT Q$
0250 FL=1
0260 X=LEN(Q$)
0270 IF X=2 OR X=4 OR X=6 THEN 290
0280 PRINT@201," ":GOTO 240
0290 FOR I=1 TO 6
0300 C(I)=0: NEXT I
0310 FOR J=1 TO X
0320 C(J)=ASC(MID$(Q$,J,1))
0330 JJ=INT((J+1)/2)
0340 REM TEST VALIDITY OF LOCATOR
0350 IF C(J)<CB(JJ) OR C(J)>CT(JJ) THEN 280
0360 C(J)=C(J)-CB(JJ)
0370 NEXT J
0380 REM CALCULATE LATITUDE/LONGITUDE
0390 T(1,2)=-
90+C(2)★10+C(4)+C(6)/24+CM(X/2)/2
0400 T(2,2)=-
180+C(1)★20+C(3)★2+C(5)/12+CM(X/2)
0401 FOR I=1 TO 2
0402 IF T(I,2)<0 THEN H%(I)=2 ELSE H%(I)=1
0403 H%(I)=1
0404 T=ABS(T(I,2))
0405 L(I,1)=INT(T)
0406 L(I,3)=(T-L(I,1))★60
0407 L(I,2)=INT(L(I,3))
0408 L(I,3)=INT((L(I,3)-L(I,2))★60+0.5)
0409 NEXT I
0410 GOTO 585
0420 PRINT@192,"TARGET LAT/LONG"
0430 K=2 0440 GOSUB 1030
0450 REM FORM TARGET LOCATOR
0460 FOR J=1 TO 2
0470 TG(J)=T(J,2)+90★J
0490 IF TG(J)=180★J THEN TG(J)=TG(J)-0.0001
0500 FOR K=3 TO 7 STEP 2
0510 M(K-J)=INT(TG(J)/(J★CM((K-1)/2)))
0520 TG(J)=TG(J)-M(K-J)★J★CM((K-1)/2)
0530 NEXT K
0540 NEXT J
0550 Q$=""
0560 FOR I=1 TO 6
0570 Q$=Q$+CHR$(M(I)+CB(INT((I+1)/2)))
0580 NEXT I
0585 GOSUB 860
0590 REM CALCULATE BEARING AND DISTANCE
0600 P=T(2,1)-T(2,2)
0610 PS=1
0620 IF P<0 THEN PS=0
0630 P=ABS(P)
0640 PM=0
0650 IF P>180 THEN PM=1
0660 E=57.29578
0670 PI=3.141592654
0680 P=P/E
0690 PA=(90-T(1,1))/E
0700 PB=(90-T(1,2))/E
0710 ZZ=COS(P)★SIN(PA)★SIN(PB)+COS(PA)★COS(PB)
0720 GOSUB 1250
0730 AB=AC
0740 SK=INT(6366.707★AB+0.5)
0750 LK=40000-SK
0760 ZZ=(COS(PB)-
COS(PA)★COS(AB))/(SIN(PA)★SIN(AB))
0770 GOSUB 1250
0780 A=AC★E
0790 A=ABS(360★(PS-PM)²-A)
0800 A1=INT(A)
0810 A2=INT((A-A1)★60+0.5)
0820 B=180+A
0830 IF B>=360 THEN B=B-360
0840 B1=INT(B)
0850 B2=INT((B-B1)★60+0.5)
0855 GOTO 920
0860 REM PRINT RESULTS
```

```
0870 CLS
0880 PRINT"TARGET: ";T$
0885 IF LEN(C$)>0 THEN PRINT@17,C$
0890 PRINT@64,"LAT:
";L(1,1);"D";L(1,2);"M";L(1,3);"S ";
0895 PRINT@86,F$(H%(1))
0900 PRINT
@96,"LONG:";L(2,1);"D";L(2,2);"M";L(2,3);"S ";
0905 PRINT@118,G$(H%(2))
0910 PRINT"LOCATOR ",Q$
0915 RETURN
0920 PRINT@224,"SHORT PATH:
BEARING";A1;"D";A2;"M"
0930 PRINT" DISTANCE";SK;" KMS"
0940 PRINT"LONG PATH: BEARING";B1;"D";B2;"M"
0950 PRINT" DISTANCE";LK;" KMS"
0960 IF FL=0 THEN 1000
0970 PRINT"LAT,LONG, BEARINGS AND DISTANCES
ONLY"
0980 PRINT"APPROXIMATE BECAUSE LAT AND
LONG"
0990 PRINT"CALCULATED FROM LOCATOR"
1000 PRINT@480,"ENTER 'Y' FOR ANOTHER
TARGET";:INPUT Y$
1010 IF Y$="Y" THEN 170
1020 GOTO 1270
1030 REM INPUT LATITUDE/LONGITUDE
1035 S(1)=0: S(2)=0
1040 PRINT@224,"LATITUDE? DEGS";:INPUT L(1,1)
1041 INPUT" MINS";L(1,2)
1042 INPUT" SECS";L(1,3)
1043 INPUT" N/S ";N$(1)
1050 IF N$(1)<>"N" THEN 1070
1060 S(1)=1: GOTO 1080
1070 IF N$(1)="S" THEN S(1)=-1
1080 INPUT"LONGITUDE? DEGS";L(2,1)
1081 INPUT" MINS";L(2,2)
1082 INPUT" SECS";L(2,3)
1083 INPUT" E/W ";N$(2)
1090 IF N$(2)<>"E" THEN 1110
1100 S(2)=1: GOTO 1120
1110 IF N$(2)="W" THEN S(2)=-1
1120 FOR I=1 TO 2
1130 IF S(I)=0 THEN 1160
1132 H%(I)=1
1134 IF S(I)<0 THEN H%(I)=2
1140 T=90+(I-1)★90
1150 IF L(I,1)>=0 AND L(I,1)<=T THEN 1180
1160  PRINT"ERROR:";L(I,1);"D";L(I,2);"M";L(I,3);"S
";N$(I)
1170 PRINT "TRY AGAIN"
1172 FOR V=1 TO 1500
1174 NEXT V
1175 PRINT@224,W$
1176 FOR V=1 TO 7
1177 PRINT W$
1178 NEXT V
1179 GOTO 1030
1180 FOR J=2 TO 3
1190 IF L(I,J)<0 OR L(I,J)>60 THEN 1160
1200 NEXT J
1210 T(I,K)=L(I,1)+L(I,2)/60+L(I,3)/3600
1220 T(I,K)=T(I,K)★S(I)
1230 NEXT I
1240 RETURN
1250 AC=-ATN(ZZ/SQR(1-ZZ★ZZ))+PI/2
1260 RETURN
1270 END
```

5 of 5.